Prof. Dr. Peter Bürgisser
Kathlén Kohn

Technische Universität Berlin
Fakultät II, Institut für Mathematik
Sekretariat MA 3-2

Exercises for Algebraic Geometry 1

Winter term 2017/2018

Exercise sheet 4

Mit Lösungsansätzen von Kathlén Kohn

Exercise 1. Let X be an affine variety, $Y \subseteq X$ closed, and $J \subseteq A(X)$ an ideal. We define the vanishing ideal $I_X(Y) := \{f \in A(X) \mid \forall p \in Y : f(p) = 0\}$ and the zero set $Z_X(J) := \{p \in X \mid \forall f \in J : f(p) = 0\}$. Show that $Z_X(I_X(Y)) = Y$ and that $I_X(Z_X(J)) = \sqrt{J}$.

Macaulay2

Exercise 2 (Exercise 1 on Sheet 2). Let $I = \langle x^2 - yz, xz - x\rangle$ and $X := Z(I) \subseteq \mathbb{A}^3$. Use `Macaulay2` to verify that I is radical and to compute the prime ideals of the irreducible components of X.

Solution.
```
QQ[x,y,z]
I = ideal(x^2-y*z, x*z-x)
dim I, degree I   --this shows that X is a curve of degree 4
radical(I) == I   --this shows that I is radical
Components = decompose(I)   --X has 3 irreducible components
Components / dim, Components / degree
--hence X consists of 2 lines and a quadratic curve
```

Exercise 3 (Exercise 3 on Sheet 3). In $\mathbb{C}^4$ with coordinates x, y, z, t, let X be the union of the two planes

$$Z(x, y) \qquad \text{and} \qquad Z(z, x - t).$$

(1) Find the vanishing ideal $I := I(X) \subset \mathbb{C}[x, y, z, t]$ with `Macaulay2`.

(2) For any $a \in \mathbb{C}$, let $I_a \subset \mathbb{C}[x, y, z]$ be the ideal obtained by substituting $t = a$ in I, and let $X_a = Z(I_a) \subset \mathbb{A}^3$.
Compute with `Macaulay2` the prime ideals of the irreducible components of X_1 and X_0, and see that X_1 is two skew lines, whereas X_0 is two lines intersecting at the origin.

(3) Verify with `Macaulay2` that I_1 is radical but that I_0 is not. Compute the radical ideal of I_0.

(4) Why is it enough to consider $a = 1$ to deduce that all ideals I_a for $a \neq 0$ are radical?

Solution.
```
R = QQ[x,y,z,t]
J1 = ideal(x,y)
J2 = ideal(z,x-t)
I = intersect(J1,J2)
--or use the radical of the product:
I = radical(J1*J2)
decompose I   --in this way, we can get the 2 planes back

sub0 = {t => 0}
sub1 = {t => 1}
I0 = sub(I,sub0)
I1 = sub(I,sub1)
decompose I0   --2 lines meeting at origin
decompose I1   --2 skew lines

radical(I1) == I1    --I1 is radical!
rad0 = radical(I0)
rad0 == I0   --I0 is not radical!
```

Let $a \in \mathbb{C}$ be non-zero. Then we have that

$$\begin{aligned} I_a &= \langle xz,\ yz,\ x(x-a),\ y(x-a)\rangle \\ &= \left\langle \frac{1}{a}xz,\ yz,\ \frac{1}{a^2}x(x-a),\ \frac{1}{a}y(x-a)\right\rangle \\ &= \left\langle \frac{x}{a}z,\ yz,\ \frac{x}{a}\left(\frac{x}{a}-1\right),\ y\left(\frac{x}{a}-1\right)\right\rangle. \end{aligned}$$

Using the change or coordinates $\tilde{x} := \frac{x}{a}$, we get that

$$I_a = \langle \tilde{x}z,\ yz,\ \tilde{x}(\tilde{x}-1),\ y(\tilde{x}-1)\rangle.$$

Hence, I_a is radical if and only if I_1 is radical. ∎

Exercise 4 (Exercise 1 on Sheet 1). Consider the following curve in $\mathbb{C}^3$:

$$C := \{(t^3, t^4, t^5) \mid t \in \mathbb{C}\} = \{(x,y,z) \in \mathbb{C}^3 \mid x^3 = yz,\ y^2 = xz,\ z^2 = x^2y\}.$$

Verify with `Macaulay2` that one needs indeed three equations to define C.
Helpful command: mingens

Solution.
```
QQ[x,y,z]
I = ideal(x^3-y*z, y^2-x*z, z^2-x^2*y)
dim I, degree I   --this shows that C is a curve of degree 5
mingens I   --this shows that there is no smaller set of defining equations
```

Exercise 5 (Exercise 2 on Sheet 1). Consider the set

$$X := \{ \begin{pmatrix} m_{00} & m_{01} & m_{02} \\ m_{10} & m_{11} & m_{12} \end{pmatrix} \in \mathbb{C}^{2\times 3} \mid m_{00}m_{11} = m_{10}m_{01}, m_{00}m_{12} = m_{10}m_{02}, m_{01}m_{12} = m_{11}m_{02} \}$$

of all 2×3-matrices of rank at most 1. Verify with Macaulay2 that X has dimension four and that one needs indeed three equations to define X.

Solution.
```
R = QQ[m_(0,0)..m_(1,2)]
M = matrix{{m_(0,0), m_(0,1), m_(0,2)},{m_(1,0), m_(1,1), m_(1,2)}}
-- or use this alternative version with shorter code:
M = transpose genericMatrix (R,3,2)
I = minors (2,M)
dim I, degree I   --this shows that X has dimension 4 and degree 3
mingens I   --this shows that there is no smaller set of defining equations
```

Exercise 6 (Related to Exercise 3 on Sheet 1). Consider the cubic surface $S \subseteq \mathbb{R}^3$ defined by

$$f = 81(x^3 + y^3 + z^3) - 189(x^2y + x^2z + xy^2 + xz^2 + y^2z + yz^2) + 54xyz + 126(xy + xz + yz) - 9(x^2 + y^2 + z^2) - 9(x + y + z) + 1.$$

Verify with Macaulay2 that there are 27 **real** lines on S and compute them explicitly! How many of these lines are defined over $\mathbb{Q}$?

Solution.
```
R = QQ[x,y,z,a,b,c,d][s]
f = 81*(x^3+y^3+z^3)-189*(x^2*y+x^2*z+x*y^2+x*z^2+y^2*z+y*z^2)+54*x*y*z
    +126*(x*y+x*z+y*z)-9*(x^2+y^2+z^2)-9*(x+y+z)+1
sub1 = {x => s, y => s*a+(1-s)*c, z => s*b+(1-s)*d}
sub2 = {x => c, y => s, z => s*b+(1-s)*d}
sub3 = {x => c, y => d, z => s}
g1 = sub(f,sub1)
(M,C1) = coefficients g1
I1 = ideal flatten entries C1
```

```
S1 = QQ[a,b,c,d]
I1 = sub(I1,S1)
dim I1, degree I1   --this yields 22 solutions

g2 = sub(f,sub2)
(M,C2) = coefficients g2
I2 = ideal flatten entries C2
S2 = QQ[b,c,d]
I2 = sub(I2,S2)
dim I2, degree I2   --this yields 5 solutions

g3 = sub(f,sub3)
(M,C3) = coefficients g3
I3 = ideal flatten entries C3
S3 = QQ[b,d]
I3 = sub(I3,S3)
dim I3, degree I3   --this yields 0 solutions

sol2 = decompose I2   --explicit solutions for 5 lines
#sol2   --I2 has been really decomposed in 5 points
sol2 / degree   --check degree of all ideals in sol2

sol1 = decompose I1
#sol1   --I1 can be only decomposed in 16 prime ideals over Q
sol1 / degree   --12 lines still come in pairs
--these lines are not defined over Q
Pairs = apply(6, i -> sol1#(i+10))
--we extract the one and only quadratic equation in each pair
Quadratics = Pairs / mingens / entries / flatten / last
S = QQ[d]
Quadratics = apply (Quadratics, q -> sub(q,S))
--we compute the discriminants of the quadratic equations
Discriminants = apply (Quadratics, q
   -> (coefficient(d,q))^2-4*coefficient(d^2,q)*coefficient(d^0,q))
--since these are all positive, every pair yields 2 real lines
```

Hence, 15 lines are defined over $\mathbb{Q}$, 12 lines only over $\mathbb{R}$. We can write down explicit solutions in terms of square roots by simply using the p-q-formula. ■